Welfare Analysis using Dynare

Tsung-Hsien Li
Academia Sinica

May 2026

Outline

01 **The Framework** — model structure, neoclassical example

02 **Perturbation & Why Order Matters** — certainty equivalence

03 **Welfare Concepts** — recursive welfare, cost λ, un- vs. conditional

04 **Dynare Implementation** — the `example.mod` file, welfare results

05 **Conclusion** — applications, tips, takeaways

References

- Schmitt-Grohé and Uribe's paper and lecture notes
- Eric Sims' lecture notes
- Dynare: Adjemian et al. (2011) — *Reference Manual*
- Johannes Pfeifer's collection of Dynare models

The Framework

For many DSGE models

$$\mathbb{E}_t f(y_{t+1}, y_t, x_{t+1}, x_t) = 0 \tag{1}$$

- x_t: predetermined (state); y_t: non-predetermined (control)
- $x_t = [x_t^1; x_t^2]'$ where x_t^1 endogenous and x_t^2 exogenous
- $x_{t+1}^2 = \tilde{h}(x_t^2, \sigma) + \bar{\eta}\sigma\epsilon_{t+1}$

Example—Neoclassical model with fixed labor

- Consider

$$\max_{C_t, K_{t+1}} \mathbb{E}_0 \sum_{t=0}^{\infty} \beta^t \frac{C_t^{1-\gamma}}{1-\gamma} \tag{2}$$

$$\text{s.t. } C_t + K_{t+1} = A_t K_t^{\alpha} + (1-\delta) K_t \tag{3}$$

$$\ln A_{t+1} = \rho \ln A_t + \eta_A \epsilon_{t+1} \tag{4}$$

- FOC

$$C_t^{-\gamma} = \beta \mathbb{E}_t C_{t+1}^{-\gamma} \left(\alpha A_{t+1} K_{t+1}^{\alpha-1} + 1 - \delta \right) \tag{5}$$

(1)'s representation

- Let $y_t = C_t$, $x_t^1 = K_t$, $x_t^2 = \ln A_t$; so, $x_t = [K_t, \ln A_t]'$
- It follows that

$$\begin{aligned}
&\mathbb{E}_t f(y_{t+1}, y_t, x_{t+1}, x_t) = 0 \\
&= \mathbb{E}_t \begin{bmatrix} y_{1,t}^{-\gamma} - \beta y_{1,t+1}^{-\gamma} \left(\alpha \cdot \exp(x_{2,t+1}) \cdot x_{1,t+1}^{\alpha-1} + 1 - \delta \right) \\ y_{1,t} + x_{1,t+1} - \exp(x_{2,t}) \cdot x_{1,t}^{\alpha} - (1-\delta) x_{1,t} \\ x_{2,t+1} - \rho x_{2,t} \end{bmatrix}
\end{aligned} \tag{6}$$

Perturbation & Why Order Matters

Perturbation

- Key idea is to interpret the solution as a function of x_t and σ

$$y_t = g(x_t, \sigma) \tag{7}$$
$$x_{t+1} = h(x_t, \sigma) + \eta\sigma\epsilon_{t+1} \tag{8}$$

- A local approximation that is valid in the neighborhood of given $(\tilde{x}, \tilde{\sigma})$
- Focusing on non-stochastic steady state $(\bar{x}, 0)$
- Taylor expansion of F, g, h at $(\bar{x}, 0)$ plus some linear algebra
- Dynare does this automatically via `stoch_simul(order=k)`

First-, second-order, or beyond?

- First-order: the certainty equivalence principle holds ($g_\sigma = h_\sigma = 0$)
- Policy functions independent of $\Sigma_\epsilon \rightarrow$ No effect of uncertainty
  - Cannot study welfare! Why?
    Any policies with same steady state yield same welfare
- We need at least a second-order approximation to study welfare
- For higher-order, it might be helpful but needs pruning and is unclear

What changes at second order?

- **Precautionary savings** — agents respond to risk by saving more
- **Jensen's inequality at work** — $\mathbb{E}[u(C)] \neq u(\mathbb{E}[C])$ when u is concave
- **Risk-adjusted steady state $\neq$ deterministic steady state** — the stochastic mean differs from $\bar{x}$, and this difference depends on Σ_ϵ
- **Policy ranking becomes possible** — two regimes with the same steady state can yield different welfare as they generate different volatilities

Welfare Concepts

Welfare measure

- Welfare defined as the present discounted value of lifetime utility

$$V_t = \mathbb{E}_t \sum_{j=0}^{\infty} \beta^j u(C_{t+j}) = u(C_t) + \beta \mathbb{E}_t V_{t+1} \tag{9}$$

- This makes V_t an additional **control variable** in the system
- Thus, $y_t = [C_t, V_t]'$ so $V_t = g^V(x_t, \sigma)$

(Un)conditional welfare

- Non-stochastic welfare (no shocks): $\overline{V}$
  - Steady-state value without shocks, deterministic
  - $\overline{V} = u(\overline{C})/(1-\beta)$
- Unconditional welfare: $\mathbb{E}V$
  - Average welfare with shocks, stochastic mean of V
- Conditional welfare: $\mathbb{E}_{ss}V$
  - Conditional on being in the steady state, but know that the model is stochastic so that shocks may happen in the future (transition)
  - Need to adjust the deterministic mean with $\frac{1}{2}g^{V}_{\sigma\sigma}\sigma^2$

Computing the welfare cost λ

- λ: fraction of consumption an agent would pay to move from old to new regime (consumption equivalence)

$$V^{\text{old}} = \mathbb{E}_0 \sum_{t=0}^{\infty} \beta^t u\big((1-\lambda)\, C_t^{\text{new}}\big) \tag{10}$$

- For CRRA utility with $\gamma \neq 1$:

$$\lambda = 1 - \left(\frac{V^{\text{old}}}{V^{\text{new}}}\right)^{\frac{1}{1-\gamma}} \tag{11}$$

- Cost of business cycles (Lucas calculation)

Which welfare measure to report?

Scenario	Measure	Rationale
Deterministic benchmark	$\overline{V}$	Welfare without shocks; measuring the cost of uncertainty
Long-run regimes	$\mathbb{E}V$	Long-run average welfare; agents have been living under the policy
One-time policy reform	$\mathbb{E}_{ss}V$	Agents start at the old steady state; transition dynamics matter
Consumption-based	λ	λ translates the welfare gap into a consumption-equivalent cost

Dynare Implementation

The example

- Recall equilibrium conditions

$$C_t^{-\gamma} = \beta \mathbb{E}_t C_{t+1}^{-\gamma} \left(\alpha A_{t+1} K_{t+1}^{\alpha-1} + 1 - \delta \right) \tag{12}$$

$$C_t + K_{t+1} = A_t K_t^{\alpha} + (1-\delta) K_t \tag{13}$$

$$\ln A_{t+1} = \rho \ln A_t + \eta_A \epsilon_{t+1} \tag{14}$$

- To study welfare, add

$$\boldsymbol{V_t = u(C_t) + \beta \mathbb{E}_t V_{t+1}} \tag{15}$$

The example.mod file

```
var k c a V;
varexo e;
parameters gamma alpha beta delta rho eta;

gamma = 2;
alpha = 1/3;
beta = 0.99;
delta = 0.025;
rho = 0.95;
eta = 0.01;
```

The `example.mod` file (cont'd)

```
model;
c^(-gamma) = beta*c(+1)^(-gamma)*(alpha*exp(a(+1))*k^(
    alpha-1)+(1-delta));
c+k = exp(a)*k(-1)^alpha+(1-delta)*k(-1);
a = rho*a(-1)+e;
V = c^(1-gamma)/(1-gamma)+beta*V(+1);
end;
```

Why in levels, not in logs? V is usually negative!

The example.mod file (cont'd)

```
steady_state_model;
k = (alpha/(1/beta-(1-delta)))^(1/(1-alpha));
c = k^alpha-delta*k;
a = 0;
V = (c^(1-gamma)/(1-gamma))/(1-beta);
end;

shocks;
var e = eta^2;
end;

stoch_simul(order=1, irf=120);
```

Experiment: How does η affect welfare?

- A nice working example to showcase what I just introduced

- How does shock volatility affect welfare?
- $\eta \in \{0.01, 0.02, 0.03, 0.04, 0.05\}$
- Three welfare measures at first-order vs. second-order
- Shock specification matters at second- and higher-order

Two Handy Things

- In the calling script, loop over the search space of η in the outer loop, and pass each η candidate into the `.mod` file with

```
set_param_value('eta',eta);
```

- Look up welfare V with

```
welfare_index = strmatch('V',M_.endo_names,'exact');
```

Welfare in Dynare

- Non-stochastic welfare (no shocks): $\overline{V}$
  - `oo_.dr.ys(welfare_index)`
- Unconditional welfare: $\mathbb{E}V$
  - `oo_.mean(welfare_index)`
- Conditional welfare: $\mathbb{E}_{ss}V$
  - `oo_.dr.ys(welfare_index)`
    `+ 0.5 * oo_.dr.ghs2(oo_.dr.inv_order_var(welfare_index))`
- Live demo with `run.m`, `ex_unadjusted.m`, and `ex_adjusted.m`

Welfare results of varying η_A

- First-order: All welfare measures are the same
- Second-order: How to cook in exogenous processes matters!
- Larger shock variance, larger difference between $\mathbb{E}V$ and $\mathbb{E}_{ss}V$

	$\eta = 0.01$	$\eta = 0.02$	$\eta = 0.03$	$\eta = 0.04$	$\eta = 0.05$
$\overline{V}$					
$\mathbb{E}V$					
$\mathbb{E}_{ss}V$					

Conclusion

Applications

- Monetary policy (Taylor rule)

$$R_t = R_{t-1}^{\rho_R} \left[\overline{R} \left(\frac{\pi_t}{\overline{\pi}} \right)^{\rho_\pi} \left(\frac{Y_t}{\overline{Y}} \right)^{\rho_y} \right]^{(1-\rho_R)} e^{\epsilon_t^R} \tag{16}$$

- Macroprudential policy

$$LTV_t = LTV_{t-1}^{\rho_l} \left[\overline{LTV} \left(\frac{B_t}{\overline{B}} \right)^{\rho_B} \left(\frac{q_t}{\overline{q}} \right)^{\rho_q} \right]^{(1-\rho_l)} e^{\epsilon_t^l} \tag{17}$$

- Housing transaction or property tax

$$C_t + K_{t+1} + q_t H_{t+1} = A_t K_t^\alpha + (1-\delta) K_t + (1-\tau) q_t H_t \tag{18}$$

Some tips

- How you specify exogenous processes matters — η vs. $\mathbb{E}[A]$
- Use `set_param_value` to loop over parameters
- Use `strmatch` to look up variable indices
- Welfare weights for two agents (like normalization)

$$W = (1-\beta_L)V^L + (1-\beta_H)V^H = u(C^L) + u(C^H) \tag{19}$$

- Use Dynare to linearize Phillips curve to keep higher-order effects
  - Need the recursive representation
  - c.f. Johannes Pfeifer's `Gali_2015_chapter_3_nonlinear.mod`

Takeaways

- Welfare analysis requires at least a second-order approximation
- Understand $\overline{V}$, $\mathbb{E}V$, and $\mathbb{E}_{ss}V$ and their computations
- For variance-based optimal rules, consider `osr` (appendix)

Appendix

Optimal simple rules (osr)

- Dynare can optimize policy coefficients to minimize a loss function:

$$\min_{\rho_\pi, \rho_y} \mathrm{Var}(\pi) + \lambda_y \mathrm{Var}(y) \tag{20}$$

```
osr_params rho_pi rho_y;

optim_weights;
  pi 1;       % weight on inflation variance
  y  0.5;     % weight on output gap variance
end;

osr(order=2);
```